This is a program to compute the colored Alexander polynomial following Ohtsuki. In what follows, n is the dimension of the representation (color), $q = \exp\left(2\pi\sqrt{-1}\,/n\right)$, and λ is a complex parameter.

First we define the quantum integer [n], note that by definition [0]=1

```
quantum[n_] := If[n === 0, 1, (q^(n/2) - q^(-n/2))/(q^(1/2) - q^(-1/2))
  ];
```

And the quantum factorial

```
qfact[n_] := Product[quantum[i], {i, 0, n}];
```

Next we define the action of E, here m is the power of E, k is the basis vector and n is the dimension of the representation

```
EE[m_, k_, n_] := Product[quantum[n - k + i + λ], {i, 1, m}];
```

Now the action of F

```
FF[m_, k_, n_] := Product[quantum[k + i], {i, 0, m - 1}];
```

Now the action of the R matrix

```
Rp[i_, j_, k_, l_, n_] :=
  Which[i + j ≠ k + l, 0, i < l, 0, True,
   (m = i - l;
    If[m > Min[k - 1, n - l, n - 1], 0,
     q^((n-1) (n-1+λ)/2 - (n-1+λ)^2/4 + (n-2 k+2 m+1+λ) (n-2 l-2 m+1+λ)/4) q^(m (m-1)/4)/qfact[m] (q^(1/2) - q^(-1/2))^m FF[m, l, n] EE[m, k, n]]
  ];
```

We also write the matrix form of Rp for our convenience

```
RpMatrix[n_] :=
 Apply[Rp, Flatten[Table[Flatten[Table[{i, j, k, l, n}, {k, 1, n}, {l, 1, n}], 1],
     {i, 1, n}, {j, 1, n}], 1], {2}] // Simplify
```

Now the action of negative R matrix

```
Rm[i_, j_, k_, l_, n_] :=
  Which[i + j ≠ k + l, 0, i > l, 0, True,
   (m = l - i;
    If[m > Min[l - 1, n - k, n - 1], 0,
     q^(-(n-1) (n-1+λ)/2 + (n-1+λ)^2/4 - (n-2 k+1+λ) (n-2 l+1+λ)/4) q^(-m (m-1)/4)/qfact[m] (q^(-1/2) - q^(1/2))^m EE[m, l, n] FF[m, k, n]])
  ];
```

And the matrix form of Rm for our convenience

```
RmMatrix[n_] :=
 Apply[Rm, Flatten[Table[Flatten[Table[{i, j, k, l, n}, {k, 1, n}, {l, 1, n}], 1],
     {i, 1, n}, {j, 1, n}], 1], {2}] // Simplify
```

Now we are ready to write our main program, here n is the color of the representation

```
ColouredAlexander[Knot_, n_] := Module[{m},
  m = Max[List @@@ List @@ Knot];
  Sum[Knot /. {
        Xp[i_, j_, k_, l_] :> Rp[s_i, s_j, s_k, s_l, n],
        Xm[i_, j_, k_, l_] :> Rm[s_i, s_j, s_k, s_l, n],
        Cu[k_] :> q^(-(1-n) (n-2 s_k+1+λ)/2),
        Ca[k_] :> q^((1-n) (n-2 s_k+1+λ)/2)
       }, ##] & @@ ({s_#, 1, n} & /@ Range[m]) / n // Simplify
 ]
```

Now let’s compute a few knots. Like the Alexander polynomial, we need to keep one component open. Hence, we present our knot as the closure of a braid, except we leave the first strand unclosed, we label the first strand by 1 and continue from top to bottom and we end with the label 1.

The trefoil, color=3

```
ColouredAlexander[Xp[4, 1, 2, 5] Xp[2, 5, 6, 3] Xp[6, 3, 4, 1] Cu[4], 3]
```

$$\frac{1}{3}\left(-5+\frac{1}{q^3}+\frac{2}{q^2}-\frac{2}{q}-6\,q-7\,q^2-2\,q^3+q^4+q^{-5-2\lambda}-3\,q^{-1-2\lambda}-q^{1-2\lambda}+q^{4-2\lambda}-q^{-4-\lambda}-q^{-3-\lambda}+q^{-2-\lambda}+4\,q^{-1-\lambda}+7\,q^{1-\lambda}+2\,q^{2-\lambda}-q^{3-\lambda}-q^{-2+\lambda}-2\,q^{-1+\lambda}-3\,q^{-2\lambda}+7\,q^{-\lambda}-q^{\lambda}+q^{2\lambda}-q^{-2(1+\lambda)}+4\,q^{1+\lambda}+3\,q^{2+\lambda}+2\,q^{3+\lambda}+q^{4+\lambda}+2\,q^{1+2\lambda}-q^{2+2\lambda}-2\,q^{3+2\lambda}-q^{2+3\lambda}-q^{3+3\lambda}+q^{5+3\lambda}+q^{6+3\lambda}+q^{4+4\lambda}+q^{6+4\lambda}+q^{8+4\lambda}\right)$$

The figure eight knot, n=3

```
ColouredAlexander[
 Ca[4] Ca[7] Xp[1, 4, 5, 2] Xm[2, 7, 8, 3] Xp[5, 8, 1, 6] Xm[6, 3, 4, 7], 3]
```

$$\frac{1}{3} q^{-2\lambda} \left(107 + \frac{4}{q^4} + \frac{6}{q^3} + \frac{31}{q^2} + \frac{66}{q} + 104 q + 76 q^2 + 36 q^3 + 11 q^4 + q^{-10-4\lambda} + q^{-6-4\lambda} - 2 q^{-8-3\lambda} - 2 q^{-7-3\lambda} - 6 q^{-5-3\lambda} - 8 q^{-4-3\lambda} - 2 q^{-2-3\lambda} + 4 q^{-7-2\lambda} + 10 q^{-5-2\lambda} + 31 q^{-3-2\lambda} + 21 q^{-1-2\lambda} + 6 q^{1-2\lambda} + q^{2-2\lambda} - 4 q^{-6-\lambda} - 3 q^{-5-\lambda} - 14 q^{-4-\lambda} - 33 q^{-3-\lambda} - 61 q^{-2-\lambda} - 74 q^{-1-\lambda} - 46 q^{1-\lambda} - 22 q^{2-\lambda} - 8 q^{3-\lambda} - q^{-3+\lambda} - 2 q^{-2+\lambda} - 12 q^{-1+\lambda} + 11 q^{-2\lambda} - 65 q^{-\lambda} - 39 q^{\lambda} + 3 q^{2\lambda} - 4 q^{-3(1+\lambda)} + 28 q^{-2(1+\lambda)} - 83 q^{1+\lambda} + q^{-4(2+\lambda)} - 4 q^{-3(2+\lambda)} + 20 q^{-2(2+\lambda)} - 97 q^{2+\lambda} - 2 q^{-3(3+\lambda)} + 4 q^{-2(3+\lambda)} - 64 q^{3+\lambda} + 2 q^{-2(4+\lambda)} - 28 q^{4+\lambda} - 4 q^{5+\lambda} + q^{-1+2\lambda} + 10 q^{1+2\lambda} + 31 q^{2+2\lambda} + 52 q^{3+2\lambda} + 31 q^{4+2\lambda} + 9 q^{5+2\lambda} + q^{6+2\lambda} - 2 q^{2+3\lambda} - 3 q^{3+3\lambda} - 13 q^{4+3\lambda} - 11 q^{5+3\lambda} - q^{6+3\lambda} + 3 q^{6+4\lambda}\right)$$

The Hopf link, n=3

```
ColouredAlexander[Ca[3] Xp[1, 3, 4, 2] Xp[4, 2, 1, 3], 3]
```

$$\frac{1}{3} q^{-4-3\lambda} \left(1 + q^4 + 2 q^5 + q^6 + q^{10} - q^{1+\lambda} - q^{2+\lambda} + q^{3+\lambda} + q^{4+\lambda} - 2 q^{5+\lambda} - 3 q^{6+\lambda} - 2 q^{7+\lambda} + q^{8+\lambda} + 2 q^{2+2\lambda} + 2 q^{3+2\lambda} - q^{4+2\lambda} - 2 q^{5+2\lambda} + 2 q^{6+2\lambda} + 2 q^{7+2\lambda} + q^{8+2\lambda} - 2 q^{4+3\lambda} - q^{5+3\lambda} + 2 q^{6+3\lambda} + q^{7+3\lambda} + 3 q^{6+4\lambda}\right)$$

The (5,2) torus knot, n=3

```
ColouredAlexander[
 Ca[6] Xp[1, 6, 7, 2] Xp[7, 2, 3, 8] Xp[3, 8, 9, 4] Xp[9, 4, 5, 10] Xp[5, 10, 1, 6], 3]
```

$$\frac{1}{3} q^{-7-6\lambda} \left(1 - q^7 - 5 q^8 - 10 q^9 - 10 q^{10} - 5 q^{11} - q^{12} + q^{19} - q^{1+\lambda} - q^{2+\lambda} - q^{5(2+\lambda)} + 2 q^{6(2+\lambda)} - q^{7(2+\lambda)} + q^{3+\lambda} - 21 q^{4(3+\lambda)} + 10 q^{5(3+\lambda)} + q^{4+\lambda} + 67 q^{3(4+\lambda)} - 2 q^{4(4+\lambda)} - 38 q^{2(5+\lambda)} + 3 q^{3(5+\lambda)} - 64 q^{2(6+\lambda)} - 3 q^{2(7+\lambda)} + 5 q^{8+\lambda} + 21 q^{9+\lambda} + 38 q^{10+\lambda} + 37 q^{11+\lambda} + 18 q^{12+\lambda} + 2 q^{13+\lambda} - q^{14+\lambda} + q^{2+2\lambda} + 2 q^{3+2\lambda} - 2 q^{4+2\lambda} - 2 q^{5+2\lambda} + q^{6+2\lambda} - 10 q^{9+2\lambda} - 67 q^{11+2\lambda} - 31 q^{13+2\lambda} + 3 q^{15+2\lambda} - q^{3+3\lambda} - 2 q^{4+3\lambda} + q^{5+3\lambda} + 3 q^{6+3\lambda} - q^{8+3\lambda} + 10 q^{10+3\lambda} + 38 q^{11+3\lambda} + 64 q^{13+3\lambda} + 31 q^{14+3\lambda} - 3 q^{16+3\lambda} + q^{4+4\lambda} + 2 q^{5+4\lambda} - 4 q^{7+4\lambda} - q^{8+4\lambda} + 2 q^{9+4\lambda} - 5 q^{11+4\lambda} - 38 q^{13+4\lambda} - 37 q^{14+4\lambda} - 18 q^{15+4\lambda} + q^{17+4\lambda} - 2 q^{5+5\lambda} - 2 q^{6+5\lambda} - q^{7+5\lambda} + 3 q^{8+5\lambda} + 2 q^{9+5\lambda} - q^{11+5\lambda} + q^{12+5\lambda} + 5 q^{13+5\lambda} + 10 q^{14+5\lambda} + 5 q^{16+5\lambda} + q^{17+5\lambda} + 2 q^{7+6\lambda} + 2 q^{8+6\lambda} - q^{9+6\lambda} - 4 q^{10+6\lambda} - q^{11+6\lambda} - 2 q^{9+7\lambda} - 2 q^{10+7\lambda} + 2 q^{11+7\lambda} + 3 q^{12+7\lambda} + 2 q^{11+8\lambda} + 2 q^{12+8\lambda} - 3 q^{13+8\lambda} - 2 q^{14+8\lambda} + q^{15+8\lambda} - 2 q^{13+9\lambda} - q^{14+9\lambda} + 2 q^{15+9\lambda} + q^{16+9\lambda} + 3 q^{15+10\lambda}\right)$$